PRÁCTICA 2: Cálculo Diferencial de Funciones de una variable

Límites y Continuidad

Uno de los primeros conceptos que se presentan en un curso de Cálculo es el de continuidad. Este concepto está íntimamente ligado al concepto de límite.
En este capítulo veremos cómo usar *Maxima* para resolver algunos problemas relacionados con todos esto.

Límites

El cálculo de límites se realiza con la orden `limit`. Con ella podemos calcular límites de funciones o de sucesiones en un número, en $+\infty$ o en $-\infty$. También podemos usar el menú **Análisis→Calcular límite**. Ahí podemos escoger, además de a qué función le estamos calculando el límite, a qué tiende la variable incluyendo los valores "especiales" como π, e o infinito. Además de esto, también podemos marcar si queremos calcular únicamente el límite por la derecha o por la izquierda.



`limit (expr,x,a)`	$\lim_{x\to a} expr$
`limit (expr,x,a,plus)`	$\lim_{x\to a^+} expr$
`limit (expr,x,a,minus)`	$\lim_{x\to a^-} expr$
`inf`	$+\infty$
`minf`	$-\infty$
`und`	indefinido
`ind`	indefinido pero acotado

El cálculo de límites con *Maxima*, como puedes ver, es sencillo. Sabe calcular límites de cocientes de polinomios en infinito

```
(%i1)limit(n/(n+1),n,inf);
(%o1) 1
```

```
(%i2)limit((x^2+3*x+1)/(2*x+3),x,-inf);
(%o2) -∞
```

aplicar las reglas de L'Hôpital,

```
(%i3)limit(sin(x)/x,x,0);
(%o3) 1
```

Incluso es capaz de dar alguna información en el caso de que no exista el límite. Por ejemplo, sabemos que las funciones periódicas, salvo las constantes, no tienen límite en ∞. La respuesta de *Maxima* cuando calculamos el límite de la función coseno en ∞ es

```
(%i1)limit(n/(n+1),n,inf);
(%o1) 1
```

Indeterminado. Este límite es equivalente a

```
(%i4)limit(cos(x),x,inf);
(%o4) ind
```

Indeterminado. Este límite es equivalente a

```
(%i5)limit(cos(1/x),x,0);
(%o5) ind
```

La función $\cos\left(\frac{1}{x}\right)$ oscila cada vez más rápidamente cuando nos acercamos al origen. Observa su gráfica.

```
(%i6)wxplot2d(cos(1/x), [x,-2,2])$
plot2d: expression evaluates to non-numeric value somewhere in plotting range.
```

(%t6)



Otro ejemplo es la función valor absoluto

```
(%i7)wxplot2d(abs(x), [x,-2,2])$
```

(%t7)




```
(%i8)limit(abs(x)/x,x,0);
(%o8) und
```

En este último límite lo que ocurre es que tenemos que estudiar los límites laterales

(%i9) limit(abs(x)/x,x,0,plus);

(%o9) 1

(%i10) limit(abs(x)/x,x,0,minus);

(%o10) -1

Por tanto, no existe el límite puesto que los límites laterales no coinciden. Si recuerdas la definición de función derivable, acabamos de comprobar que la función valor absoluto no es derivable en el origen.

Continuidad

El estudio de la continuidad de una función es inmediato una vez que sabemos calcular límites. Una función $f : A \subset \mathbb{R} \to \mathbb{R}$ es continua en $a \in A$ si $\lim_{x\to a} f(x) = f(a)$. Conocido el valor de la función en el punto, la única dificultad es, por tanto, saber si coincide o no con el valor del límite.

Con respecto a las funciones discontinuas, la gráfica puede darnos una idea del tipo de discontinuidad. Si la discontinuidad es evitable, es difícil apreciar un único pixel en la gráfica. Una discontinuidad de salto es fácilmente apreciable. Por ejemplo, la función signo, esto es, $\frac{|x|}{x}$, tiene un salto en el origen que *Maxima* une con una línea vertical.

(%i11) wxplot2d([abs(x)/x], [x,-1,1], [y,-2,2])$

plot2d: expression evaluates to non-numeric value somewhere in plotting range

(%t11)



Cálculo de derivadas

Para calcular la derivada de una función real de variable real, una vez definida, por ejemplo, como $f(x)$, se utiliza el comando `diff` que toma como argumentos la función a derivar, la variable con respecto a la cual hacerlo y, opcionalmente, el orden de derivación. **diff**

`diff(`*`expr,variable`*`)`	derivada de *expr*
`diff(`*`expr,variable,n`*`)`	derivada n-ésima de *expr*

O bien en modo paleta a través del menú: **Análisis → Diferencial**



(%i15) kill(all)$

(%i1) diff(tan(x),x);

(%o1) $\sec(x)^2$

La orden `diff` considera como constantes cualquier otra variable que aparezca en la expresión a derivar, salvo que explícitamente manifestemos que están relacionadas.

(%i2) diff(a*x*sin(a+x),x);

(%o2) $a\sin(x+a)+a\,x\cos(x+a)$

También podemos trabajar con funciones que previamente hayamos definido.

(%i3) f(x):=x^2+sin(x^2);

(%o3) $f(x):=x^2+\sin\left(x^2\right)$

(%i4) diff(f(x),x);

(%o4) $2\,x\cos\left(x^2\right)+2\,x$

La tercera entrada de la orden `diff` nos permite calcular derivadas de orden superior. Por ejemplo, la cuarta derivada de f sería la siguiente.

(%i5) diff(f(x),x,4);

(%o5) $16\,x^4\sin\left(x^2\right)-12\sin\left(x^2\right)-48\,x^2\cos\left(x^2\right)$

Reutilizar la derivada

Las derivadas sucesivas de una función nos dan mucha información sobre la función original y con frecuencia nos hace falta utilizarlas de nuevo, ya sea para calcular puntos críticos, evaluar para estudiar monotonía o extremos relativos, etc. Es por ello que es cómodo escribir la derivada como una función. Hay varias formas en las que podemos hacerlo. Nosotros vamos a utilizar la orden `define`

(%i5) define(g(x),diff(f(x),x));

(%o5) $g(x):=2\,x\cos\left(x^2\right)+2\,x$

Recta tangente y normal a una función

Ejemplo. Consideremos la función $f(x) = x^3 - 2x^2 - x + 2$ y su derivada, a la que notaremos df,

(%i7) f(x):=x^3-2*x^2-x+2;

(%o7) $f(x):=x^3-2x^2-x+2$

(%i8) define(df(x),diff(f(x),x));

(%o8) $df(x):=3x^2-4x-1$

La recta tangente a una función f en un punto a es la recta $y = f(a) + f'(a)(x - a)$. La definimos.

(%i9) tangente(x,a):=f(a)+df(a)*(x-a);

(%o9) $tangente(x,a):=f(a)+df(a)(x-a)$

Ya podemos dibujar la función y su tangente en 1:

(%i10) wxplot2d([f(x),tangente(x,1)],[x,-2,3])$

(%t10)



La recta normal es la recta perpendicular a la recta tangente. Su pendiente es $-1/f'(a)$ y, por tanto, tiene como ecuación $y = f(a) - \frac{1}{f'(a)}(x - a)$. Con todo lo que ya tenemos hecho es muy fácil dibujarla.

(%i11) normal(x,a):=f(a)-1/df(a)*(x-a);

(%o11) $normal(x,a):=f(a)-\frac{1}{df(a)}(x-a)$

(%i12) wxplot2d([f(x),normal(x,1)],[x,-2,3])$

(%t12)



En la figura anterior, la recta normal no parece perpendicular a la recta tangente. Eso es por que no hemos tenido en cuenta la escala a la que se dibujan los ejes. Prueba a cambiar la escala para que queden perpendiculares.

```
(%i14) wxplot2d([f(x),normal(x,1)], [x,-1,2],same_xy)$
```

(%t14)



Máximos y mínimos relativos

En esta sección vamos a aprender a localizar extemos relativos de una función f. Para ello encontraremos las soluciones de la ecuación de punto crítico: $f'(x) = 0$. Y para resolver dicha ecuación podemos usar el comando `solve`.

Recordemos algunos criterios para la discusión de los puntos críticos:

Criterio de la derivada primera:

Si $f'(c) = 0$ $f'(c-\Delta x) > 0$ y $f'(c+\Delta x) < 0$	$(c, f(c))$ es un **máximo local** *de* $f(x)$
Si $f'(c) = 0$ $f'(c-\Delta x) < 0$ y $f'(c+\Delta x) > 0$	$(c, f(c))$ es un **mínimo local** *de* $f(x)$

Criterio de la derivada segunda:

Si $f'(c) = 0$ y $f''(c) < 0$	$(c, f(c))$ es un **máximo local** *de* $f(x)$
Si $f'(x) = 0$ y $f''(c) > 0$	$(c, f(c))$ es un **mínimo local** *de* $f(x)$

Pero si su segunda derivada siguiese siendo nula, f´´(c) = 0, entonces no podríamos afirmar nada.

Criterio de las derivadas sucesivas:
Debemos obtener derivadas sucesivas hasta llegar a una derivada de orden n que no se anule:

Si n es:	**Signo $f^{n)}(c)$**	**el punto es un**
par	$f^{n)}(c) > 0$	mínimo local
	$f^{n)}(c) < 0$	máximo local
impar	$f^{n)}(c) \neq 0$	punto de inflexión

Ejemplo Calculemos los extremos relativos de la función $f(x) = 2x^4 + 4x$, $\forall x \in \mathbb{R}$.

Comenzamos, entonces, presentándosela al programa (no olvidéis borrar de la memoria la anterior función f) y pintando su gráfica para hacernos una idea de dónde pueden estar sus extremos.

(%i15) kill(all)$

(%i1) f(x):=2*x^4+4*x;

(%o1) $f(x):=2\,x^4+4\,x$

(%i2) wxplot2d(f(x),[x,-2,1])$

(%t2) 

Parece que hay un mínimo en las proximidades de -1. Para confirmarlo, calculamos los puntos críticos de f.

(%i3) define(d1f(x),diff(f(x),x));

(%o3) $d1f(x):=8\,x^3+4$

(%i4) puncritf:solve(d1f(x),x);

(%o4) $\left[x=-\frac{\sqrt{3}\,\%i-1}{2^{4/3}},x=\frac{\sqrt{3}\,\%i+1}{2^{4/3}},x=-\frac{1}{2^{1/3}}\right]$

Observamos que hay sólo una raíz real que es la única que nos interesa.

(%i5) puncritf[3];

(%o5) $x=-\frac{1}{2^{1/3}}$

Debemos tener cuidado con esta salida… solo nos interesa la parte derecha de la salida.
Si definimos la derivada segunda:

(%i6) define(d2f(x),diff(f(x),x,2));

(%o6) $d2f(x):=24\,x^2$

para calcular su valor en el punto crítico 3, hay muchas formas. La más cómoda quizá sea usar el operador , (coma) de sustitución que vimos en la práctica anterior;

(%i7) d2f(x),puncritf[3];

(%o7) $3\,2^{7/3}$

Otra forma es “rescatar” la parte derecha de la salida con la orden **rhs** y darle un nombre:

(%i8) a:rhs(puncritf[3]);

(%o8) $-\frac{1}{2^{1/3}}$

(%i9) d2f(a);

(%o9) $3\,2^{7/3}$

Por tanto la función f tiene un mínimo relativo en dicho punto a. ¿Puede haber otro extremo más? ¿Cómo podemos asegurarnos? ◁

Ejemplo Vamos a calcular los extremos relativos de la función:

$$f(x) = x^5 + x^4 - 11x^3 - 9x^2 + 18x + 10 \text{ en el intervalo } [-4, 4]$$

Procedemos de la misma forma que en el ejemplo anterior.

```
(%i9) f(x):=x^5+x^4-11*x^3-9*x^2+18*x+10$
(%i10) wxplot2d(f(x),[x,-4,4])$
```

(%t10)



A simple vista observamos que hay:
a) un máximo relativo entre -3 y -2, b) un mínimo relativo entre -2 y -1, c) un máximo relativo entre 0 y 1, y d) un mínimo relativo entre 2 y 3.
Vamos entonces a derivar la función e intentamos calcular los ceros de f' haciendo uso del comando `solve`.

(%i11) define(d1f(x),diff(f(x),x));

(%o11) $\mathrm{d1f}(x):=5\,x^4+4\,x^3-33\,x^2-18\,x+18$

(%i12) puncritf:solve(d1f(x),x)$

Las soluciones que nos da el programa no son nada manejables (comprobarlo!!), así que vamos a intentar resolver la ecuación *f'(x)=0* de forma numérica, en primer lugar con la ayuda del comando **algsys**.

(%i13) puncritf:algsys([d1f(x)],[x]);

(%o13) $[[x=-2.600820933812211],[x=2.363816114681167],$
$[x=-1.096856414613424],[x=0.5338613861386139]]$

Como vemos, en este ejemplo, el comando **algsys** sí es capaz de encontrar los puntos críticos. Ahora utilizamos el criterio de la derivada segunda para ver si estos puntos críticos son máximos o mínimos relativos.

```
(%i19) define(d2f(x),diff(f(x),x,2))$
       d2f(x),puncritf[1];
       d2f(x),puncritf[2];
       d2f(x),puncritf[3];
       d2f(x),puncritf[4];
(%o16) −117.0276599100646
(%o17) 157.2020930532604
(%o18) 42.43722399568166
(%o19) −46.77166062872889
```

Por tanto en el primer y cuarto puntos hay máximo relativo, mientras que en los otros dos tenemos mínimos relativos.

Nota: En otros ejercicios es posible que el comando **algsys** tampoco sea capaz de localizar los ceros de la derivada primera.

En estos ejemplos será necesario recurrir al comando **find_root**. Si lo aplicamos al ejemplo anterior la idea es apoyarse en la gráfica anterior buscando intervalos en los que es posible encontrar solución. En este caso se obtiene:

```
(%i20) puncritf:[find_root(d1f(x),x,-3,-2),find_root(d1f(x),x,-2,0),
         find_root(d1f(x),x,0,1),find_root(d1f(x),x,2,3)];
(%o20) [−2.600821117505113,−1.096856508567827,0.5338613537430753,
       2.363816272329865]
```

Polinomio de Taylor

Polinomio de Taylor de orden n de una función f en un punto a:

$$\begin{aligned} T(f,a,n)(x) &= f(a) + f'(a)(x-a) + \frac{f''(a)}{2!}(x-a)^2 \\ &+ \frac{f'''(a)}{3!}(x-a)^3 + \cdots + \frac{f^{n)}(a)}{n!}(x-a)^n \\ &= \sum_{k=1}^{n} \frac{f^{k)}(a)}{k!}(x-a)^k \end{aligned}$$



El programa tiene una orden que permite calcular directamente el polinomio de Taylor centrado en un punto a. Se trata del comando `taylor`. En concreto, el comando `taylor(f(x),x,a,n)` nos da el polinomio de Taylor de la función f centrado en a y de grado n. Haciendo uso del menú podemos acceder al comando anterior desde **Análisis→Calcular serie**. Entonces se abre una ventana de diálogo en la que, escribiendo la expresión de la función, la variable, el punto en el que desarrollamos y el orden del polinomio de Taylor, obtenemos dicho polinomio. Como en otras ventanas similares, si marcamos la casilla de `Especial`, podemos elegir π o e como centro para el cálculo del desarrollo. **taylor**

`taylor(f(x),x,a,n)`	polinomio de Taylor de la función f en el punto a y de orden n
`trunc(polinomio)`	convierte polinomio de Taylor en un polinomio

Veamos un ejemplo.

(%i20) kill(all)$

(%i1) taylor(cos(x),x,0,5);

(%o1)/T/ $1-\frac{x^2}{2}+\frac{x^4}{24}+\ldots$

En teoría, un polinomio de Taylor de orden más alto debería aproximar mejor a la función. Ya hemos visto cómo aproxima la recta tangente a la función coseno. Vamos ahora a dibujar las gráficas de la función $f(x) = \cos(x)$ y de su polinomio de Taylor de orden 8 en el cero para comprobar que la aproximación es más exacta.

(%i2) wxplot2d([cos(x),taylor(cos(x),x,0,5)],[x,-2,2])$

(%t2) 

Pero si aumentamos el dominio podemos ver que el polinomio de Taylor se separa de la función cuando nos alejamos del origen.

(%i3) wxplot2d([cos(x),taylor(cos(x),x,0,5)],[x,-6,6])$

(%t3) 

Esto es lo esperable: la función coseno está acotada y el polinomio de Taylor, como todo polinomio no constante, no lo está. Eso sí, si aumentamos el grado del polinomio de Taylor vuelven a parecerse:

(%i4) wxplot2d([cos(x),taylor(cos(x),x,0,14)],[x,-6,6])$

(%t4) 

Observación *Maxima* tiene dos formas de representar internamente los polinomios. Sin entrar en detalles, no se guardan de la misma forma un polinomio de Taylor y un polinomio cualquiera. Esto puede dar lugar a algunas sorpresas. Por ejemplo, hemos visto cómo el polinomio de Taylor nos sirve para aproximar una función, pero, en lugar de representar la función y dicho polinomio, podríamos representar la diferencia. Veamos que ocurre[9]. Definimos las funciones,

(%i1) kill(all)$

(%i1) f(x):=cos(x);

(%o1) $f(x):=\cos(x)$

(%i2) define(g(x),taylor(cos(x),x,0,5));

(%o2)/T/ $g(x):=1-\frac{x^2}{2}+\frac{x^4}{24}+\ldots$

y dibujamos la diferencia

(%i3) wxplot2d(f(x)-g(x),[x,-5,5])$

(%t3)



¿Cómo puede salir 0? ¿Es que no hay diferencia? Sí la hay. Ya lo sabemos: si evaluamos en algún punto podemos ver que el resultado no es cero.

(%i4) f(2)-g(2);

(%o4)/R/ $\frac{3\cos(2)+1}{3}$

Como hemos avisado antes, *Maxima* maneja de forma diferente un polinomio de Taylor y un polinomio "normal".

trunc La orden `trunc(polinomio de Taylor)` nos permite pasar de polinomio de Taylor a polinomio "normal". Con estos ya no tenemos este problema.

(%i5) wxplot2d(f(x)-trunc(g(x)),[x,-5,5])$

(%t5)



EJERCICIOS PROPUESTOS

1) Calcular los límites siguientes:

a) $\lim\limits_{x\to\infty} \frac{3x^4 sen^2(\frac{1}{x}) \ln(1+\frac{1}{x})}{(x+6)^{\frac{\pi x+2}{4x+1}}}$

b) $\lim\limits_{x\to 0} \frac{\cos x + e^{-\frac{x^2}{2}} - 2}{x^2}$

c) $\lim\limits_{x\to 0} (\cos x)^{\cot^2 x}$

d) $\lim\limits_{x\to 0} \frac{\sqrt[n]{1+x}-1}{x}$ $x \in \mathbb{N}$

2) Estudiar la continuidad de la función $f(x) = xLn|x|$ si $x \neq 0$ y $f(0) = 0$

3) Hallar las derivadas por la derecha y por la izquierda de la función $f(x) = |x-1|$ en el punto $x = 1$. Representarla gráficamente en el intervalo $[0, 2]$.

4) Estudiar si la siguiente función es derivable en los puntos $x = -1, x = 1, x = 0$.

$$f(x) = \begin{cases} 1 & si \quad x < -1 \\ |x| & si \quad -1 \leq x \leq 1 \\ -1 & si \quad 1 \leq x \end{cases}$$

Representar la gráfica en el intervalo $[-3, 3]$.

5) Dada la función:

$$f(x) = x^2 sen(\frac{1}{x^2}) + 3\cos x + x \quad \text{si} \quad x \neq 0 \quad \text{y} \quad f(0) = 3$$

a) Comprobar la continuidad en $x = 0$.

b) Hallar la derivada en $x = 0$, siendo ésta el límite:

$$\lim_{x\to 0} \frac{f(x) - f(0)}{x - 0}$$

c) Calcular el corte de la gráfica con el eje de abscisas (con $find\_root$) y dibujar la gráfica de f junto con la tangente en ese punto.

6) Hallar máximos, mínimo y puntos de inflexión de la función:

$$f(x) = x^3 - x^5$$

Comprobarlo mediante la representación gráfica.

7) Hallar los extremos relativos y los puntos de inflexión de la función:

$$f(x) = 10x^6 - 24x^5 + 15x^4 + 2$$

8) Dibujar en una misma ventana la función:

$$f(x) = x\cos(3x)$$

y sus polinomios de McLaurin de orden 2, 4, 6 y todos con diferentes colores.